Ребят подскажите пожалуйста где ошибка,данный запрос в консоле запросов отрабатывает нормально,выводя нормальный резьтат.
когда вставляю в процедуру при отладке,в выборке выдает ошибки значений
Подскажите пожалуйста,еще очень не опытен,может что то упустил..
Еще если не трудно подскажите плиз мне нужно заполнить макет печатной формы, что бы все поля сгруппировались
по полю "ТипПрибора", уменя что то не совсем это получается

```1c
    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    | ОперацииИМатериалы.Материал1 КАК Материал1,
    | ОперацииИМатериалы.Материал2 КАК Материал2,
    | ОперацииИМатериалы.Операция КАК Операция,
    | ПланДеталиОстатки.План1Остаток КАК План1Остаток,
    | ПланДеталиОстатки.План2Остаток КАК План2Остаток,
    | ПланДеталиОстатки.Номенклатура.ДецимальныйНомер КАК НоменклатураДецимальныйНомер,
    | ПланДеталиОстатки.Номенклатура КАК Номенклатура,
    | ПланДеталиОстатки.Номенклатура.ПринадлежностьКТипуПрибора КАК НоменклатураПринадлежностьКТипуПрибора
    |ИЗ
    | РегистрНакопления.ПланДетали.Остатки КАК ПланДеталиОстатки
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ОперацииИМатериалы КАК ОперацииИМатериалы
    | ПО ПланДеталиОстатки.Номенклатура.Ссылка = ОперацииИМатериалы.Номенклатура.Ссылка
    |ГДЕ
    | ОперацииИМатериалы.Операция.Ссылка = &Операция
    |ИТОГИ ПО
    | НоменклатураПринадлежностьКТипуПрибора";

    Запрос.УстановитьПараметр("Операция", справочники.ВидыОперацийМеханическийУчасток.НайтиПоНаименованию("Пластмасса"));

    ВыборкаТип = Запрос.Выполнить().Выбрать();
```
---
Необходимо спозиционироваться на элементе выборки с помощью Выборка.Следующий()

```1c
    .........
    ВыборкаТип = Запрос.Выполнить().Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам);

    Пока ВыборкаТип.Следующий() Цикл
    ...........
    ВыборкаДетальныеЗаписи = ВыборкаТип.Выбрать();

    Пока ВыборкаДетальныеЗаписи.Следующий() Цикл
    .............
    КонецЦикла;
    ............
    КонецЦикла;
```
---
когда вставляю в процедуру при отладке,в выборке выдает ошибки значений
исправление ошибок стоит начинать с внимательного чтения сообщения об ошибке. Иногда (не всегда) там бывает существенная информация.
---
Необходимо спозиционироваться на элементе выборки с помощью Выборка.Следующий()

```1c
    .........
    ВыборкаТип = Запрос.Выполнить().Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам);

    Пока ВыборкаТип.Следующий() Цикл
    ...........
    ВыборкаДетальныеЗаписи = ВыборкаТип.Выбрать();

    Пока ВыборкаДетальныеЗаписи.Следующий() Цикл
    .............
    КонецЦикла;
    ............
    КонецЦикла;
```
---
Спасибо, все получилось, в принципе все заполняет как нужно.
Но в ВыборкеТип в локальных переменных почему то пишет все равно "ошибка чтения значения"
пока все норм работает, но не возникнут ли в дальнейшем из за этого какие то проблемы?
---
5 команд и два запроса. Жуть.

```1c
    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    | ОперацииИМатериалы.Материал1 КАК Материал1,
    | ОперацииИМатериалы.Материал2 КАК Материал2,
    | ОперацииИМатериалы.Операция КАК Операция,
    | ПланДеталиОстатки.План1Остаток КАК План1Остаток,
    | ПланДеталиОстатки.План2Остаток КАК План2Остаток,
    | ПланДеталиОстатки.Номенклатура.ДецимальныйНомер КАК НоменклатураДецимальныйНомер,
    | ПланДеталиОстатки.Номенклатура КАК Номенклатура,
    | ПланДеталиОстатки.Номенклатура.ПринадлежностьКТипуПрибора КАК НоменклатураПринадлежностьКТипуПрибора
    |ИЗ
    | РегистрНакопления.ПланДетали.Остатки КАК ПланДеталиОстатки
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ОперацииИМатериалы КАК ОперацииИМатериалы
    | ПО ПланДеталиОстатки.Номенклатура = ОперацииИМатериалы.Номенклатура
    | Левое соединение ВидыОперацийМеханическийУчасток КАК ВидыОперацийМеханическийУчасток
    | по ОперацииИМатериалы.Операция = ВидыОперацийМеханическийУчасток.ссылка
    |ГДЕ
    | ВидыОперацийМеханическийУчасток.Наименование = &Операция
    |ИТОГИ ПО
    | НоменклатураПринадлежностьКТипуПрибора";

    Запрос.УстановитьПараметр("Операция", "Пластмасса");

    ВыборкаТип = Запрос.Выполнить().Выбрать();
```
---
Тогда уж так:

```1c
    | ...............
    | ВНУТРЕННЕЕ соединение ВидыОперацийМеханическийУчасток КАК ВидыОперацийМеханическийУчасток
    | по ОперацииИМатериалы.Операция = ВидыОперацийМеханическийУчасток.ссылка
    | И ВидыОперацийМеханическийУчасток.Наименование = "Пластмасса"
    | И НЕ ВидыОперацийМеханическийУчасток.ЭтоГруппа
    |
    |ИТОГИ ПО
    | НоменклатураПринадлежностьКТипуПрибора
```
---
Согласен, лучше так

ВНУТРЕННЕЕ соединение ВидыОперацийМеханическийУчасток КАК ВидыОперацийМеханическийУчасток
 | по ОперацииИМатериалы.Операция = ВидыОперацийМеханическийУчасток.ссылка
 | И ВидыОперацийМеханическийУчасток.Наименование = &Операция

В постах ТС ничего про группу не писано. Кроме того, ТС наименование параметром передаёт. Может это будет входящий параметр процедуры.
---
Ребят спасибо за помощь, сейчас отошел, через час опробою, спс еще раз
---
ПО ПланДеталиОстатки.Номенклатура.Ссылка = ОперацииИМатериалы.Номенклатура.Ссылка
ОперацииИМатериалы.Операция.Ссылка = &Операция

Не надо так делать.
---
Не итогами надо пользоваться а группировкой по полю "ТипПрибора".
Ошибки, пожалуйста, опубликуйте, которые получаете.
---
можно и итогами - все зависит от того, что необходимо сделать с результатом запроса.
---
ошибка значений когда сиотрю в локальнвх переменных,по выборке,а так не пишет ошибки,может что то я не понял.
---
еще раз:
необходимо спозиционироваться на элементе выборки методом Следующий().
Без позиционирования выборка вам будет выдавать то, что вы видите.
---
```1c
    Запрос.УстановитьПараметр("Операция",Поиск);

    ВыборкаТип = Запрос.Выполнить().Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам);

    ШапкаТаблицы = Макет.ПолучитьОбласть("ШапкаТаблицы");
    ТипПрибор = Макет.ПолучитьОбласть("Тип");
    Таблица = Макет.ПолучитьОбласть("Таблица");
    ТабДок.Вывести(ШапкаТаблицы);

    Пока ВыборкаТип.Следующий() Цикл
    ТипПрибор.Параметры.ТипПрибора = ВыборкаТип.НоменклатураПринадлежностьКТипуПрибора;
    ТабДок.Вывести(ТипПрибор);
    ВыборкаДетальныеЗаписи = ВыборкаТип.Выбрать();

    Пока ВыборкаДетальныеЗаписи.Следующий() Цикл
    Таблица.Параметры.Заполнить(ВыборкаДетальныеЗаписи);
    ТабДок.Вывести(Таблица);
    КонецЦикла;

    КонецЦикла;
```

Вашей подсказкой запоняется все норм. Но не понимаю почему если на печать заполняет все данные верно, что за проблема в локальных переменных с "ошибкой чтения значения"
---
Но не понимаю почему если на печать заполняет все данные верно, что за проблема в локальных переменных с "ошибкой чтения значения"

Потому что значения выборки ВыборкаТип надо смотреть внутри цикла.